
In this Maple file, we check that the formulas found for the 

second element of the Painlevé 2 hierarchy corresponds to the 

theoretical ones.
Loading the results after symplectic reduction

restart:

with(LinearAlgebra):

tinfty14:=1;

tinfty24:=-1;

tinfty13:=0;

tinfty23:=0;

tinfty12:=1/2*tau2;

tinfty22:=-tinfty12;

tinfty11:=1/2*tau1;

tinfty21:=-tinfty11;

tinfty20:=-tinfty10;

P1:=lambda->0:

tdP2:=unapply(-lambda^6-tau2*lambda^4-tau1*lambda^3-(2*

tinfty10+1/4*tau2^2)*lambda^2,lambda);

P2:=unapply(-lambda^6-tau2*lambda^4-tau1*lambda^3-(2*

tinfty10+1/4*tau2^2)*lambda^2+Pinfty02+Pinfty12*lambda,lambda);

dtdP2dlambda:=unapply(diff(tdP2(lambda),lambda),lambda);

tdP2onlambda:=unapply(tdP2(lambda)/lambda,lambda):

Definition of the matrices necessary to define the vector (c_{\infty,0}, c_{\infty,1}) and verification 
that we recover the correct formulas

Q:=Matrix(2,2,0):

Q[1,1]:=1:

Q[1,2]:=q1:

Q[2,1]:=1:

Q[2,2]:=q2:

RHS1:=Matrix(2,1,0):

RHS1[1,1]:=p1^2+tdP2(q1)+h*(p2-p1)/(q1-q2)+h*q1^2:

RHS1[2,1]:=p2^2+tdP2(q2)+h*(p1-p2)/(q2-q1)+h*q2^2:

Q;

Q^(-1);

RHS1;

simplify(1/(q1-q2)*h*(p2-p1)/(q1-q2)-1/(q1-q2)*h*(p1-p2)/(q2-q1))

;

simplify(-q2/(q1-q2)*h*(p2-p1)/(q1-q2)+q1/(q1-q2)*h*(p1-p2)/(q2-

q1));

CVector:=simplify(Multiply(Q^(-1),RHS1)):

Cinfty0:=CVector[1,1]:

Cinfty1:=CVector[2,1]:

Cinfty1bis:=(p1^2-p2^2)/(q1-q2)-( (q1^2+q2^2+1/2*tau2)^2-q1^2*

q2^2 +(2*tinfty10-h))*(q1+q2)-(q1^2+q1*q2+q2^2)*tau1;

factor(Cinfty1-Cinfty1bis);

Cinfty0bis:=-(p1^2-p2^2)*q1/(q1-q2) -h*(p1-p2)/(q1-q2)+p1^2 +

(q1^4+q2^4+q1^3*q2+q2^3*q1+q1^2*q2^2+1/4*tau2^2+ (q1^2+q1*q2+

q2^2)*tau2+(q1+q2)*tau1+ 2*tinfty10-h)*q1*q2:

Cinfty0bis:=(p2^2*q1-q2*p1^2)/(q1-q2) -h*(p1-p2)/(q1-q2) +(q1^4+

q2^4+q1^3*q2+q2^3*q1+q1^2*q2^2+1/4*tau2^2+ (q1^2+q1*q2+q2^2)*

tau2+(q1+q2)*tau1+ 2*tinfty10-h)*q1*q2;

Cinfty0ter:=-(p1^2-p2^2)*q1/(q1-q2) -h*(p1-p2)/(q1-q2)+p1^2 +(

(q1^5-q2^5)/(q1-q2)+1/4*tau2^2+ (q1^3-q2^3)/(q1-q2)*tau2+(q1^2-

q2^2)/(q1-q2)*tau1+ 2*tinfty10-h)*q1*q2:

Cinfty0ter:=-(p1^2-p2^2)*q1/(q1-q2) -h*(p1-p2)/(q1-q2)+p1^2 -

(tdP2onlambda(q1)-tdP2onlambda(q2))/(q1-q2)*q1*q2 -h*q1*q2:

factor(Cinfty0-Cinfty0bis);

factor(Cinfty0-Cinfty0ter);

Definition of the matrix M_{\infty} and of the vectors \nu_{\infty,1} and \nu_{\infty,2}

Minfty:=Matrix(4,4,0):

Minfty[1,1]:=2:

Minfty[2,2]:=2:

Minfty[3,3]:=2:

Minfty[4,4]:=2:

Minfty[3,1]:=tau2:

Minfty[4,1]:=tau1:

Minfty[4,2]:=tau2:

Minfty;

Minfty^(-1);

E3:=Matrix(4,1,0):

E3[3,1]:=1:

E4:=Matrix(4,1,0):

E4[4,1]:=1:

E3;

E4;

Multiply(Minfty^(-1),E3);

Multiply(Minfty^(-1),E4);

nu2infty1:=1/2*Multiply(Minfty^(-1),E3)[3,1];

nu2infty2:=1/2*Multiply(Minfty^(-1),E3)[4,1]; 

nu1infty1:=Multiply(Minfty^(-1),E4)[3,1];

nu1infty2:=Multiply(Minfty^(-1),E4)[4,1];

nu1inftyVector:=Matrix(2,1,0):

nu1inftyVector[1,1]:=nu1infty1:

nu1inftyVector[2,1]:=nu1infty2:

nu1inftyVector;

nu2inftyVector:=Matrix(2,1,0):

nu2inftyVector[1,1]:=nu2infty1:

nu2inftyVector[2,1]:=nu2infty2:

nu2inftyVector;

Definition of the matrix V and the vector \mu_{\tau_1} and \mu_{\tau_2}

V:=Matrix(2,2,0):

V[1,1]:=1:

V[1,2]:=1:

V[2,1]:=q1:

V[2,2]:=q2:

V;

V^(-1);

Mu1inftyVector:=Multiply(V^(-1),nu1inftyVector);

Mu2inftyVector:=Multiply(V^(-1),nu2inftyVector);

mutau11:=Mu1inftyVector[1,1];

mutau12:=Mu1inftyVector[2,1];

mutau21:=Mu2inftyVector[1,1];

mutau22:=Mu2inftyVector[2,1];

Definition of the theoretical Hamiltonian evolutions using the formulas proved in the article. 

hdq1dtau1:=2*p1*mutau11- h*(mutau12+mutau11)/(q1-q2);

hdq2dtau1:=2*p2*mutau12- h*(mutau12+mutau11)/(q2-q1);

hdp1dtau1:=simplify(h*(mutau12+mutau11)*(p2-p1)/(q1-q2)

^2+mutau11*(-diff(tdP2(q1),q1)+Cinfty1-2*h*q1));

hdp2dtau1:=simplify(h*(mutau12+mutau11)*(p1-p2)/(q1-q2)^2+

mutau12*(-diff(tdP2(q2),q2)+Cinfty1-2*h*q2));

Ham1:=factor(-h*(mutau12+mutau11)*(p1-p2)/(q1-q2)+mutau11*(p1^2+

tdP2(q1)+h*q1^2)+mutau12*(p2^2+tdP2(q2)+h*q2^2)) ;

hdq1dtau2:=simplify(2*p1*mutau21- h*(mutau22+mutau21)/(q1-q2));

hdq2dtau2:=simplify(2*p2*mutau22- h*(mutau22+mutau21)/(q2-q1));

hdp1dtau2:=simplify(h*(mutau22+mutau21)*(p2-p1)/(q1-q2)^2+

mutau21*(-diff(tdP2(q1),q1)+Cinfty1-2*h*q1));

hdp2dtau2:=simplify(h*(mutau22+mutau21)*(p1-p2)/(q1-q2)^2+

mutau22*(-diff(tdP2(q2),q2)+Cinfty1-2*h*q2));

Ham2:=factor(-h*(mutau22+mutau21)*(p1-p2)/(q1-q2)+mutau21*(p1^2+

tdP2(q1)+h*q1^2)+mutau22*(p2^2+tdP2(q2)+h*q2^2)) ;

Comparing the theoretical formulas giving the Hamiltonian evolutions with the one obtained from 
solving the compatibility equations (in a previous Maple file)

simplify(mutau12+mutau11);

simplify(mutau22+mutau21);

hdp1dtau1bis:=(p1^2-p2^2)/2/(q1-q2)^2+1/2*(5*q1^4+4*q1^3*q2+3*

q1^2*q2^2+2*q1*q2^3+q2^4)

+(q1+q2/2)*tau1+1/2*(3*q1^2+2*q1*q2+q2^2)*tau2+tau2^2/8+tinfty10-

h/2;

factor(series(hdp1dtau1-hdp1dtau1bis,p1=0));

hdp2dtau1bis:=-(p1^2-p2^2)/2/(q1-q2)^2+1/2*(5*q2^4+4*q2^3*q1+3*

q2^2*q1^2+2*q2*q1^3+q1^4)

+(q2+q1/2)*tau1+1/2*(3*q2^2+2*q2*q1+q1^2)*tau2+tau2^2/8+tinfty10-

h/2;

factor(series(hdp2dtau1-hdp2dtau1bis,tau2=0));

Ham1bis:=(p1^2-p2^2)/2/(q1-q2)-(1/2)*(q1+q2)*((q1^2+q2^2)^2-q1^2*

q2^2)

-1/2*(q1^2+q1*q2+q2^2)*tau1-1/2*(q1+q2)*(q1^2+q2^2)*tau2

-1/8*(q1+q2)*tau2^2 -1/2*(q1+q2)*(2*tinfty10-h);

factor(series(Ham1-Ham1bis,tau2=0));

hdp1dtau2bis:=-(1/4)*q2*(p1^2-p2^2)/(q1-q2)^2 -(1/4)*h*(p1-p2)/

(q1-q2)^2

-(1/4)*q2*(5*q1^4+4*q1^3*q2+3*q1^2*q2^2+2*q1*q2^3+q2^4) 

-(1/4)*q2*(2*q1+q2)*tau1 -1/4*q2*(3*q1^2+2*q1*q2+q2^2)*tau2  

-1/16*q2*tau2^2

-1/4*q2*(2*tinfty10-h);

factor(series(hdp1dtau2-hdp1dtau2bis,tinfty10=0));

dp1dtau2ter:=-(p1-p2)*(p1*q2+p2*q2+h)/(4*(q1-q2)^2)

-(1/4)*q2*(2*q1+q2)*tau1-(1/4)*q2*(3*q1^2+2*q1*q2+q2^2)*tau2-

(1/16)*q2*tau2^2-(1/4)*q2*(5*q1^4+4*q1^3*q2+3*q1^2*q2^2+2*q1*

q2^3+q2^4)-(1/2)*q2*(tinfty10-h/2);

dp2dtau2ter:=(p1-p2)*(p1*q1+p2*q1+h)/(4*(q1-q2)^2)-(1/4)*q1*

(q1+2*q2)*tau1

-(1/4)*q1*(q1^2+2*q1*q2+3*q2^2)*tau2-(1/16)*q1*tau2^2-(1/4)*q1*

(q1^4+2*q1^3*q2+3*q1^2*q2^2+4*q1*q2^3+5*q2^4)

-(1/2)*q1*(tinfty10-h/2);

factor(series(hdp1dtau2-dp1dtau2ter,tinfty10=0));

factor(series(hdp2dtau2-dp2dtau2ter,tinfty10=0));

hdp2dtau2bis:=(1/4)*q1*(p1^2-p2^2)/(q1-q2)^2 +(1/4)*h*(p1-p2)/(q1

-q2)^2

-(1/4)*q1*(5*q2^4+4*q2^3*q1+3*q2^2*q1^2+2*q2*q1^3+q1^4) 

-(1/4)*q1*(2*q2+q1)*tau1 -1/4*q1*(3*q2^2+2*q2*q1+q1^2)*tau2  

-1/16*q1*tau2^2

-1/4*q1*(2*tinfty10-h);

factor(series(hdp2dtau2-hdp2dtau2bis,tinfty10=0));

Ham2bis:=(1/4)*(q1*p2^2-q2*p1^2)/(q1-q2)+1/4*(q1^4+q1^3*q2+q1^2*

q2^2+q1*q2^3+q2^4)*q2*q1

-(1/4)*h*(p1-p2)/(q1-q2)

+(1/4*(q1+q2))*q2*q1*tau1+(1/4*(q1^2+q1*q2+q2^2))*q2*q1*

tau2+1/16*q2*q1*tau2^2

+(1/4)*q2*q1*(2*tinfty10-h);

factor(series(Ham2-Ham2bis,p2=0));

